Elec 422: VLSI Design I Fall 1997

November 18, 1997 Additional Information on Ring Oscillator

Spice Simulation

Attached is a Spice simulation of three inverters connected to make a ring oscillator. After some initial transients, the period of oscillation is about 4ns. The frequency is then about 250MHz. The spice input file ringosc.spcin can be found on Owlnet in /home/cavallar/demo/1997/ringosc. Magic layout is is also in that directory.

Note that this is an example and not a production grade circuit. The first inverter to the left has a pullup of L = 2, W = 6 and a pulldown of L = 2, W = 3. The middle inverter has the same ratio. The final inverter at the right has a pullup of L = 2, W = 12 and a pulldown of L = 2, W = 6. The output feedback node loop is run in polysilicon.

Figure 1: Plot of three stage Ring Oscillator.

Elec 422: Fall 1997, Notes on Ring Oscillator 2

*** SPICE DECK created from ringosc.sim, tech=scmos
M1 1 4 5 3 CMOSP L=2.0U W=12.0U
M2 5 4 0 2 CMOSN L=2.0U W=6.0U
M3 1 6 4 3 CMOSP L=2.0U W=6.0U
M4 4 6 0 2 CMOSN L=2.0U W=3.0U
M5 1 5 6 3 CMOSP L=2.0U W=6.0U
M6 6 5 0 2 CMOSN L=2.0U W=3.0U
C7 6 0 42.0F
C8 5 0 97.0F
C9 4 0 41.0F
C10 1 0 87.0F
* GND 0
* Vdd 1
* NMOS 2
* PMOS 3
* CMOSN 2
* CMOSP 3
* out2 4
* loop 5
* out1 6
.MODEL CMOSN NMOS LEVEL=2 LD=0.240703U TOX=407.000000E-10
+ NSUB=3.133604E+15 VTO=0.711043 KP=4.879000E-05 GAMMA=0.3801
+ PHI=0.6 UO=575.027 UEXP=0.129885 UCRIT=24726.9
+ DELTA=0.82016 VMAX=63588.7 XJ=0.250000U LAMBDA=2.474695E-02
+ NFS=6.455307E+12 NEFF=1 NSS=1.000000E+10 TPG=1.000000
+ RSH=29.040000 CGDO=3.063325E-10 CGSO=3.063325E-10 CGBO=5.135289E-10
+ CJ=1.130300E-04 MJ=0.699680 CJSW=5.235300E-10 MJSW=0.263078 PB=0.800000
* Weff = Wdrawn - Delta_W
* The suggested Delta_W is -0.60 um
.MODEL CMOSP PMOS LEVEL=2 LD=0.250000U TOX=407.000000E-10
+ NSUB=6.222000E+15 VTO=-0.80612 KP=2.113000E-05 GAMMA=0.5357
+ PHI=0.6 UO=249.519 UEXP=0.226251 UCRIT=18368.3
+ DELTA=1.75689 VMAX=40684.7 XJ=0.250000U LAMBDA=5.307119E-02
+ NFS=3.574015E+11 NEFF=1.001 NSS=1.000000E+10 TPG=-1.000000
+ RSH=87.160000 CGDO=3.181644E-10 CGSO=3.181644E-10 CGBO=6.682147E-10
+ CJ=2.493100E-04 MJ=0.551790 CJSW=2.815500E-10 MJSW=0.311229 PB=0.800000
* Weff = Wdrawn - Delta_W
* The suggested Delta_W is 0.04 um
*
* From N13L SPICE LEVEL 2 PARAMETERS 1991 nwell
*
*
* Elec422 SPICE EXPERIMENT TEMPLATE;
* J. Cavallaro 11/88, updated 11/89,
* updated N.D.Hemkumar & Kota Kishore 11/90, 11/91
* ------------------------------------

Elec 422: Fall 1997, Notes on Ring Oscillator 3

* Set BASIC VOLTAGE levels
*vdd <Vdd> <GND> dc 5
vdd 1 0 dc 5
* set substrate voltages : P-sub = Vdd; N-sub = GND
*vs1 <CMOSP> <Vdd> dc 0
vs1 3 1 dc 0
*vs2 <CMOSN> <GND> dc 0
vs2 2 0 dc 0

* ------------------------------------

* Set other constant circuit inputs, for example b, cin:
* None here.

* ------------------------------------

* Set Circuit Input which will change, for example a:
* input pulse between node and GND (initially 0 ) of:
* pulse (init_value pulse_value delay rise_time fall_time pulse_width period)
** NOTE: Commented out - no input specified.
** vin <loop> <GND> 0 pulse(0 5 0ns 0.1ns 0.1ns 4.8ns 10ns)
* vin 5 0 0 pulse(0 5 0ns 0.1ns 0.1ns 4.8ns 10ns)

* ------------------------------------

* Do analysis: give increments and total time for analysis.
.tran .1ns 50ns

* If running in batch mode spice -b, then ascii plots are made
* Plot Voltages
*.plot tran v(<loop>)
.plot tran v(5)

* set hcopydevtype=postscript

.end

Figure 2: Spice output of node loop.